Chapter 4. Color, Music, and Pizzazz

Drawing thin black lines on a dull white screen gets very boring very fast, doesn’t it? So before we go any further, let’s add some pizzazz to Logo procedures.

Add color to your procedures! Add music. Animate the turtle. Change its shape. Make flashing signs. There are all sorts of things you can do in Logo — far too many to describe here.

But let’s get started anyway. You can explore with us so later you can discover the rest on your own.

Adding Some Color

You can change the color of the background, the color used to fill shapes, and the color of the turtle’s pen. You can even change the size of the turtle’s pen.
Using Primary Colors

To draw in color on the computer screen, you have to mix the three primary light colors of red, green, and blue.

What’s a primary color? Simple — it’s a color that you cannot get from mixing two or more colors together. For example, yellow is the mix of equal parts of red and green light. White is the combination of all colors. Black is the absence of light or no mix of anything.

These mixes work great with colored light. But don’t try it with paint or crayons. That’s another set of primary colors: red, blue, and yellow. Where green is a primary color light, you mix yellow and blue paints together to make green. Make sense?

Didn’t think so. So let’s just focus on light for now.

Let’s say that your color monitor displays 256 colors. That means it can mix 256 shades of red, of green, and of blue.

Now let’s put those colors in a list of numbers. Each number represents a different shade.

Red = \[255 \ 0 \ 0\] (256 Red, 0 Green, 0 Blue)

Green = \[0 \ 255 \ 0\] (0 Red, 256 Green, 0 Blue)

Blue = \[0 \ 0 \ 255\] (0 Red, 0 Green, 256 Blue)

Each of these examples says that there is all of one color, but none of the other two. This also tells you that 100% yellow, for example, can’t be made from just one color. You have to mix red and green together to get yellow.
Yellow = [255 255 0] (255 Red, 255 Green, 0 Blue)

We said that white is the mixture of all colors. So what would be the list of numbers to display white?

White = [? ? ?]

You probably know already. But let’s see if we can prove what it is supposed to be.

If white is the mixture of all colors, let’s first take a look at some of the other common colors and the mix of primary colors that you use to create them.

---

**Logo Color Commands**

MSW Logo gives you far more than just the few colors Logy is using at the beginning of this chapter. You’ve got thousands. MSW Logo color commands use a list of three numbers to mix of red, green, and blue light.

SETPENCOLOR [255 000 000] — SETPC, for short — tells Ernestine what color to use to draw a line.

SETFLOODCOLOR [0 0 0] — SETFC, for short — lets you fill a closed shape with a color. You must go inside the shape to use this command.

SETCOLOR [255 0 0] — SETSC, for short — sets the background color to whatever you want it to be.
Color, Music, and Pizzazz

**Using the Set Menu**

Of course, setting all those numbers for all those colors can be a real hassle, especially if you are not familiar with color mixing. That’s where the Set Menu comes in handy.

Open the Set menu and you’ll see three color options:

- PenColor
- FloodColor
- ScreenColor

Selecting any of these three options lets you custom mix colors to get just the color you want. Here’s the Pencolor window.

You can left-click on any of the eight choices shown across the top. You can slide each of the Red, Green, and Blue buttons (shown at the right of each color bar). The color is displayed in the larger box in the upper right corner. Or you can press one of the arrow keys at each end of the color selection bars.

Go ahead. Open the **Set** menu and select **PenColor**. Then experiment with the buttons and arrow keys. When you have the exact color you want, left-click on **OK** to set the pen to that color.
COLORS.LGO

Morf decided that writing some procedures would make life a bit easier. For example:

```
TO PINK
OP [255 200 255]
END
```

**SPECIAL NOTE:** OP is the shortcut for OUTPUT. This is a command that must be used with another procedure. It does just what the name says, it outputs something to another procedure. You’ll hear lots more about this later. For now, let’s keep it simple.

The color PINK tells you something about the colors you use with Logo. What do you think it is?

The higher the numbers in the list, the lighter the shade of the color being displayed. Make sense?

Try this:

```
SETSC [255 0 0]
```

*The screen turns bright red.*

What color do you think this is? SETSC [128 0 0] Try it and see. It’s a darker shade of red, right?

As the numbers go down, the colors get darker. When you start adding other colors to red, like you did with PINK above, the colors get even lighter.
Take a look at the colors listed below. They’re in the COLORS.LGO procedure on the diskette that came with this book. The lighter colors all have higher numbers while the darker ones have lower numbers.

TO BLACK
OP [000 000 000]
END

TO BLUE
OP [000 000 255]
END

TO BROWN
OP [128 64 0]
END

TO CYAN
OP [0 255 255]
END

TO DKGREEN
OP [0 128 128]
END

TO GREEN
OP [000 255 000]
END

TO MAGENTA
OP [255 0 255]
END
TO OLIVE
OP [128 128 0]
END

TO ORANGE
OP [255 128 255]
END

TO PURPLE
OP [128 0 128]
END

TO RED
OP [255 0 0]
END

TO YELLOW
OP [255 255 0]
END

Now make some of your own color procedures. You might also try experimenting to make white.

TO _________
OP [ _______ _______ _______ ]
END

TO _________
OP [ _______ _______ _______ ]
END

TO _________
OP [ _______ _______ _______ ]
END
Now Morf doesn’t need to remember all the different codes for colors. He simply adds the color name to his procedures.

Black and White

SETPC BLACK sets the pen color to [0 0 0] or black.

That tells you something about black, doesn’t it? Black is the absence of all color, right? So there are no color values inside those brackets for black.

Try this:

SETSCREENCOLOR [0 0 0]

If [0 0 0] is the absence of all color, what would be the combination of all colors? How about

[255 255 255]

Go ahead. Set the pen color to [255 255 255]. Then give the turtle a command to see what color line is drawn. What’s that tell you?

Since you can make any other color by combining red, green, and blue, the combination of 100% red, green, and blue must be the combination of all colors, right?

And that’s white, isn’t it? Guess we proved something, didn’t we?
A Flood of Color

SETFC BLUE sets the flood or fill color to [0 0 255] or blue. Use it with the FILL command to fill closed shapes with color.

TO FILLIT
SETPC BLACK
REPEAT 4 [FD 100 RT 90]
PU RT 45 FD 40 PD
SETFC BLUE FILL
PU HOME PD
END

The first thing this procedure does is draw a square using a black pen color. Then the turtle picks the pen up, moves inside the square, and puts the pen down. The flood color is set to blue, and then the square is filled.

________________________________________________
SPECIAL NOTE: If FILL didn’t work just right on your computer, don’t be too surprised. It works differently on different computers. The same goes for setting the screen color.

One way to fix the screen is to left-click on the Zoom menu and then select “In.” The picture gets twice as big as it was. And the square should look filled. If not, another way to fix it is to “minimize” your screen and then restore it again. You can also move the Commander window over the drawing and then put it back where it belongs.

________________________________________________

Cyan is a light blue color. Let’s set the screen to cyan, set the pen to red, and fill the square with yellow.
That’s not bad. But the red line is sorta thin.

---

**Setting the Pen Size**

Tired of skinny lines? Try colorful fat ones.

You can change the size of the lines that are drawn by using the SETPENSIZE command or by using the Set/PenSize menu option. Be sure to check that option out.

SETPENSIZE takes two inputs representing width and height. There’s a long technical reason why there are two numbers. The easy thing is to just remember that both numbers should be the same.

The standard or default values are [1 1]. You can make those values just about anything you want. Try this:

SETPENSIZE [500 500] FD 100

Now that just about fills the screen, doesn’t it? So you need to be careful not to get too carried away.
Try this:

TO FILLIT3
SETSC CYAN
SETPC RED
SETPENSIZE [5 5]
REPEAT 4 [FD 100 RT 90]
PU RT 45 FD 40 PD
SETPFC YELLOW FILL
PU HOME PD
END

There! Now you can see a bold red line.

Want to try something really new? How about this one! First, write a simple triangle procedure.

TO TRI
RT 30 REPEAT 3 [FD 100 RT 120]
END

Now let’s add some pizazz!

TO PIZZAZ
SETSC WHITE
SETPENSIZE [120 120] SETPC PURPLE TRI
SETPENSIZE [100 100] SETPC BLUE TRI
SETPENSIZE [80 80] SETPC MAGENTA TRI
SETPENSIZE [60 60] SETPC RED TRI
SETPENSIZE [40 40] SETPC ORANGE TRI
SETPENSIZE [20 20] SETPC YELLOW TRI
SETPENSIZE [5 5] SETPC GREEN TRI
END
Now that you have the pizazz, let’s add some flash!

TO FLASH
REPEAT 20 [CS SETSC BLACK PIZZAZ]
END

Wow! Not bad, huh? Of course, what this looks like depends somewhat on the type of computer you are using. A 120 MHz Pentium is a lot faster than a 25 MHz 386 system.

Try something like this, maybe

TO TRI
REPEAT 3 [FD 100 RT 120]
END

TO PIZZAZZ
SETSC CYAN CS HT
PU LT 45 FD 100 RT 45 PD
SETPENSIZE [120 120] SETPC PURPLE RT 30 TRI
PU HOME LT 45 FD 80 RT 45 PD
SETPC BLUE RT 30 TRI
PU HOME LT 45 FD 60 RT 45 PD
SETPC MAGENTA RT 30 TRI
PU HOME LT 45 FD 40 RT 45 PD
SETPC RED RT 30 TRI
PU HOME LT 45 FD 20 RT 45 PD
SETPC ORANGE RT 30 TRI
PU HOME PD
SETPC YELLOW RT 30 TRI
PU HOME RT 135 FD 20 LT 135 PD
SETPC GREEN RT 30 TRI
END
This gives you another idea or two of what you can do with the basic color commands. We’ll use these more as we go on. But for now, why not explore what you can do with color and the shapes that you know.

- Make a neon sign that flashes your name.
- Do you remember the Logo sports activities? Why not make a flashing home run or touchdown sign?
- Create a scene and color it in. Draw houses of different colors. Add lawns and flowers, a sky with clouds.
- Make a colorful map of your town or your neighborhood.

What’s the Color

“With all that flashy stuff, how can you ever remember what color you’re using?”

That’s easy. Type

SHOW PENCOLOR

_The color mix is displayed in the Commander window; for example, [0 255 0]_

You can also look in the Status window. Go ahead. Look over in the box of Commander Buttons. Left-click on the Status button.

_The Status window is displayed and the button changes to NoStatus._

The Status window tells you a lot about what’s happening on the screen, about the turtle, the pen, and other stuff. There’s a box there that tells you the pencolor, the flood color, and the screen color. Left-click on NoStatus to close the window.
Bitblock

Here’s a sneak preview of some new commands you’ll meet in the next chapter along with another command that can add pizzazz to your drawings.

The BITBLOCK command uses SETFLOODCOLOR to put a block of color on the screen. Just type

```
BITBLOCK <width><height>
```

You don’t have to draw a block and fill it. Just tell Logo what size block you want. Take a look.

Here’s the procedure that created this picture:

```
TO COLOR :A :B :C
  IF :A > 240 [STOP]
  SETFLOODCOLOR (LIST :A :B :C)
  PU FD 15 RT 90 FD 15 LT 90 PD
  BITBLOCK 100 100
  COLOR :A + 15 :B + 15 :C + 15
END
```
TO BLOCKS
PU CS HT
MAKE "A 0
MAKE "B 0
MAKE "C 0
COLOR :A :B :C
END

Hm-m-m-m! There’s the LIST command. It tells SETFLOODCOLOR to make a list from the variables :A, :B, and :C. You’ll get into Variables in the next chapter. Right now, let’s just focus on the BITBLOCK command.

Try a few blocks to see how to use BITBLOCK. First tell Logo what color to use to fill the block. Type BITBLOCK followed by the number of steps wide the block is to be and the number of steps high. Make them anything you want.

Here’s a procedure that tells you a few things about BITBLOCK. What can you discover?

TO BLOCKSQ
SETFC [255 0 0]
BITBLOCK 200 20
SETFC [0 255 0]
BITBLOCK 20 200
PU RT 90 FD 200 RT 90
SETFC [0 0 255]
BITBLOCK 20 200
PU HOME FD 200 PD
SETFC [128 128 128]
BITBLOCK 220 20
END
Color, Music, and Pizzazz

Hiding the Commander

Does the Commander window ever get in the way of your graphics, especially now as you start adding pizzazz?

Well, why not hide it? MSW Logo has an ICON command that will do just that for you. For example, ICON “COMMANDER changes the Commander window into a small icon in the lower left corner of the screen. UNICON “COMMANDER brings it back again. To see how these commands works, let’s play with the BLOCKS procedure again.

```
TO BLOCKS
  PU CS HT ICON “COMMANDER
  LT 90 FD 400 LT 90 FD 320
  MAKE "A 0
  MAKE "B 0
  MAKE "C 0
  COLOR :A :B :C
  WAIT 600
  UNICON “COMMANDER
END

TO COLOR :A :B :C
  IF :A > 240 [STOP]
  SETFLOODCOLOR (LIST :A :B :C)
  PU FD 15 RT 90 FD 15 LT 90 PD
  BITBLOCK 100 100
  COLOR :A + 7 :B + 7 :C + 7
END
```

Not bad! We threw in another command you’ll find useful, the WAIT command. This command causes Logo to pause for as many 1/60ths of a second as you select. For example, WAIT 600 means to wait 10 seconds before doing the next command.
Adding Pizzazz to Text

You can’t have a sign — even a flashing sign — without some words written on it, right? You may have already tried to draw your name using turtle commands. That’s one way to write text on the graphics screen.

Is there another way to put some fancy text in one of the turtle’s signs?

Sure is. Actually, there are a couple of ways to do it. The easiest way is to use Set/Font option. Remember? Font is one of the boxes in the Status window.

Open the Set Menu and select the Font option.

The Font window is displayed.

The three small windows list the fonts that are available on your computer, the style that is currently being used such as:

- regular like this
- italic, which means the type is at an angle list this
• bold **like this**
• **bold italic** like this.

The last column lists the type sizes that are available. The sizes are listed in points.

There are 72 points to the inch. That will give you some idea of what the size you select will look like. On the other hand, down below a sample of your selection is displayed to give you an even better idea of what it will look like.

If you want to doctor up the font so that it looks just how you want it to look, you can use the `SETTEXTFONT` command. `SETTEXTFONT` can get very complicated. That’s because of all the options it gives you. But it really isn’t so bad. Here’s how you use it.

`SETTEXTFONT [font]`

`[font]` is a list that shows the following information:

```
[[Name of the typeface]
Height
Width
Orientation
Weight
Italic
Underline
StrikeOut
CharSet
OutPrecision
ClipPrecision
Quality
PitchAndFamily].
```
“Logy, how do you expect anyone to remember all that stuff. I don’t know what most of it means.”

“I don’t either, Morf. But for those who know lots about fonts and typefaces, this is important stuff.”

Let’s try something. It’ll show you how to use SETTEXTFONT. Type

SETTEXTFONT [HELV]

Now, to add text to the graphics screen, use the LABEL command. LABEL prints its input on the screen. For example, type

LABEL "HELLO"

What happened?

There’s the word HELLO reading up and down. That’s because the turtle is facing toward the top of the screen. Try this one:

CS RT 90 LABEL "HELLO"

Look better? You can write text at any angle you want.

Now let’s take a look at all those options. To see what options Logo using, type

SHOW TEXTFONT

The TEXTFONT options are displayed:

[[Helv] 143 67 0 400 0 0 0 2 0 2 34]
Now you can start comparing these with the list of options on the previous page. Try this as an example:

```
[[Helv] 72 67 0 900 1 1 0 0 2 0 2 34]
```

What changes did this example make? You can read all about SETTEXTFONT and TEXTFONT in the MSW Logo On-line Help file.

Why not go ahead and explore some other possible changes? If yours look a bit crazy, just type CS and no one will ever know.

After that, how about some sound?

---

**Add Some Sound Effects**

MSW Logo allows you to do lots of different things with sounds including music. You also have a number of multimedia commands available for controlling sound cards, CD-ROM players, and more. The MCI commands let you add sounds to your procedures. You can even narrate your own Logo show in your own voice if your computer lets you do
that. But that’s a big subject for a whole other book. We’ll just deal with the Logo commands for now, the one’s that let you make sounds that will play through the speakers on your PC.

SOUND is the command for making Logo music. It takes a list of two inputs: the frequency of the sound and the length of time it is to play. The length of time, or duration, is based on the speed of your computer. So you’ll have to do some experimenting to make it sound the way you want.

Frequency is the number of sound waves or sound vibrations per second.

So what’s a sound wave you ask?

When you drop a pebble into a pool of water, you see the waves move out from where the pebble hit the water. Sound acts the same way. It travels in waves out in all directions from the source of the sound.

Think of that line through the middle as the surface of pond when there are no waves. When you make a wave, part of it rises above the level of the pond. Part of it sinks below the level of the pond.
When you make lots of sound waves in each second, the sound is higher — fewer waves per second gives a lower sound. For example, when a piano wire vibrates 262 times each second, you get the note Middle C. When it vibrates 440 times each second, you get the note A above middle C. At 220 vibrations per second, you get the note A below C.

Here are two octaves of sounds you can use to play music.

<table>
<thead>
<tr>
<th>FREQUENCY</th>
<th>NOTE</th>
</tr>
</thead>
<tbody>
<tr>
<td>220</td>
<td>A (below Middle C)</td>
</tr>
<tr>
<td>223</td>
<td>A# (the # stands for Sharp)</td>
</tr>
<tr>
<td>247</td>
<td>B</td>
</tr>
<tr>
<td>262</td>
<td>C (Middle C)</td>
</tr>
<tr>
<td>277</td>
<td>C#</td>
</tr>
<tr>
<td>294</td>
<td>D</td>
</tr>
<tr>
<td>311</td>
<td>D#</td>
</tr>
<tr>
<td>330</td>
<td>E</td>
</tr>
<tr>
<td>349</td>
<td>F</td>
</tr>
<tr>
<td>370</td>
<td>F#</td>
</tr>
<tr>
<td>392</td>
<td>G</td>
</tr>
<tr>
<td>415</td>
<td>G#</td>
</tr>
<tr>
<td>440</td>
<td>A</td>
</tr>
<tr>
<td>466</td>
<td>A#</td>
</tr>
<tr>
<td>494</td>
<td>B</td>
</tr>
<tr>
<td>523</td>
<td>C</td>
</tr>
<tr>
<td>554</td>
<td>C#</td>
</tr>
<tr>
<td>587</td>
<td>D</td>
</tr>
<tr>
<td>622</td>
<td>D#</td>
</tr>
<tr>
<td>659</td>
<td>E</td>
</tr>
<tr>
<td>698</td>
<td>F</td>
</tr>
</tbody>
</table>
Making Music

Time to put all this good sound stuff to use. First let’s write procedures for the basic scale, Middle C to High C (CC).

TO C
SOUND [262 100 ]
END

TO D
SOUND [294 100 ]
END

TO E
SOUND [330 100 ]
END

TO F
SOUND [349 100 ]
END

TO G
SOUND [392 100 ]
END

TO A
SOUND [440 100 ]
END
Now you can start playing music from the keyboard. Just press the letter for the note you want to hear and press ENTER.

There’s another way also. You can write a song procedure.

TO SAINTS
SOUND [262 50 ]
SOUND [330 50 ]
SOUND [349 50 ]
SOUND [392 150 ]
WAIT 50
SOUND [262 50 ]
SOUND [330 50 ]
SOUND [349 50 ]
SOUND [392 50 ]
SOUND [392 150 ]
WAIT 50
SOUND [262 50 ]
SOUND [330 50 ]
SOUND [349 50 ]
SOUND [392 100 ]
SOUND [330 100 ]
SOUND [330 150 ]
SOUND [262 100 ]
SOUND [262 100 ]
SOUND [330 100 ]
SOUND [294 100 ]
END
SAINTS is a song popular in New Orleans. It’s part of “When the Saints Go Marching In.”

Rabbit Trail 15. Musical Fractions

Do you have trouble with fractions? Well, don’t feel lonely. Everyone has trouble with fractions. Thousands of years ago when people first started to think about numbers, they had trouble with fractions.

A few years ago, we learned about a neat way to understand fractions from Judi Harris, O.L.E. That’s short for Official Logo Enthusiast. She wrote an article for a Logo newspaper that is included on the diskette that comes with this book. It’s called MUSIC.TXT. Teachers and parents will be particularly interested in this article.

As Judi says, people have had an easy way to deal with fractions from the very beginning, ever since they began to play music. So let’s use her example. Only we’ll show it to you in MSW Logo.

TO ROW
SOUND [262 40 ]
SOUND [262 40 ]
SOUND [262 30 ]
SOUND [294 10 ]
SOUND [330 40 ]
SOUND [330 30 ]
SOUND [294 10 ]
SOUND [330 30 ]
SOUND [349 10 ]
SOUND [392 80 ]
END
This exercise is a lot easier if you draw the actual musical notes on music paper. If you don’t have actual music paper, draw five lines on a piece of paper. Or maybe you can just used regular lined paper.

The first note in the song procedure is called a quarter note. The last note is called a half note. If you look at the length (duration) numbers, an interesting idea begins to appear. Look at the lengths of the first and last notes:

\[ 40 + 40 = 80 \]

Strangely enough, one-quarter plus one-quarter equals one-half. What else can you think of?

\[ \frac{1}{4} + \frac{1}{4} = \frac{1}{2} \]

\[ 25 \text{ cents} + 25 \text{ cents} = 50 \text{ cents} \]

What about the other notes and their lengths? The length of 10 is one-quarter of the length of the quarter note. What does that equal?

\[ \frac{1}{4} \times \frac{1}{4} = \frac{1}{16} \]

A sixteenth note is a quarter note with two flags flying from the top. An eighth note has one flag and a quarter note has none. When you play two sixteenth notes close together, the two flags stretch from one note to the next.

The fourth note in ROW (SOUND [262 30]) would be written as an eighth note with a period next to it. The period is musical shorthand, meaning to add half the length of an eighth note or a sixteenth note, to the eighth note. If that sounds confusing, look at the lengths in the Logo procedure.
If a length of 10 is one-quarter that of a quarter note, then a length of 30 should be three-quarters that of the quarter note:

$$3/4 \times 1/4 = 3/16$$

When you have an eighth note linked with a period followed by a sixteenth note, this is the same as a 3/16th and a 1/16th note played together.

What is this equal to?

$$3/16 + 1/16 = 4/16 = 1/4$$

In one simple ten-note phrase, you have played with all sorts of things that have to do with fractions. But there’s a lot more in Judi’s article. Check it out!

Some More Ideas

You’ve added color and sound to your procedures. Before you go too far, why not see what you can do with these new commands.

What can you do with shapes now? Can you color them?

Why not draw a colorful birthday cake and then play Happy Birthday? How about drawing colorful lines that streak across the screen, like a rock concert laser show? Remember the FLASH procedure? That gives you an idea on how to get started.

Here’s a problem everyone gets caught in once in a while. Someone gives you a great set of procedures but you can’t figure out how to start it.
Wouldn’t it be great if procedures could start by themselves when they’re loaded?

Take a look.

Starting Procedures Automatically

There are a couple of different ways to start procedures without typing the commands.

An easy way is simply to add instructions in the Editor window like this:

```
TO HUT
REPEAT 4 [FD 60 RT 90]
FD 60 RT 30 FD 60 RT 120
REPEAT 20 [BK 6 FD 66 BK 60 RT 3]
PU HOME PD
END

HT PU LT 90 FD 200 RT 90 PD
HUT
PU LT 90 FD 120 RT 90 PD
HUT
PU LT 90 FD 40 RT 90 PD
HUT
PU RT 90 FD 40 LT 90 PD
HUT
PU RT 90 FD 120 LT 90 PD
HUT
PU RT 90 FD 200 LT 90 PD
HUT
```

When you load the HUT procedure, you’ll get a row of huts across the screen.
You can also use STARTUP. This gives you an introduction to the variables you’ll meet in the next chapter. You want to

MAKE “STARTUP [HUTS]

TO HUT
REPEAT 4 [FD 60 RT 90]
FD 60 RT 30 FD 60 RT 120
REPEAT 20 [BK 6 FD 66 BK 60 RT 3]
PU HOME PD
END

TO HUTS
HT PU LT 90 FD 200 RT 90 PD
HUT
PU LT 90 FD 120 RT 90 PD
HUT
PU LT 90 FD 40 RT 90 PD
HUT
PU RT 90 FD 40 LT 90 PD
HUT
PU RT 90 FD 120 LT 90 PD
HUT
PU RT 90 FD 200 LT 90 PD
HUT
END

When you load any MSW Logo procedure, the first thing that Ernestine does is search for the STARTUP variable. If she find it, she does what the variable is set to do. In this case, you said
MAKE the STARTUP variable [HUTS]

which tells Ernestine to run the HUTS procedure as soon as it is loaded into your workspace.